Estructura de un programa normal

La característica más destacada de un programa normal, es que su estructura permite la utilización de funciones definidas por el usuario fácilmente.  Para que sea más cómodo, las funciones definidas por el usuario son colocadas en archivos separados con la extensión “.mqh” y son almacenadas en la ruta “C:\Archivos de Programa\Carpeta MetaTrader\experts\include\”
Por lo general, un programa normal utiliza las tres funciones especiales, y estas llaman a las funciones definidas por el usuario. A su vez, las funciones definidas por el usuario, también pueden llamar a otras funciones definidas por el usuario, donde cada función, cumple una tarea específica y limitada dentro del programa. 

Un asesor experto, puede contener funciones definidas por el usuario con las más diversas características. Algunas funciones, por ejemplo, pueden dedicarse al seguimiento de eventos y gestión de datos, otras funciones se pueden dedicar a abrir y cerrar posiciones, unas terceras funciones se pueden dedicar a hacer  cálculos, por ejemplo la lógica de la estrategia, o calcular el costo de las órdenes, etc.  La decisión sobre que funciones utilizar, depende de la finalidad que tenga el asesor experto y que utilidad quiera darle el usuario.  En la figura 155, se puede ver el diagrama de la estructura de un asesor experto o EA normal, donde se utilizan funciones definidas por el usuario. 

Fig. 155. Diagrama de la estructura de un programa normal (asesor experto).

Las flechas en el diagrama muestran las relaciones entre las funciones. Por ejemplo, la función que lleva la contabilidad de las ordenes en el EA, se llama desde las funciones especiales init() y start() pero también se puede llamar desde otra parte del programa. A la derecha del diagrama, se ven las flechas que conectan entre sí a las funciones definidas por el usuario. Por ejemplo, la función que contiene la lógica de la estrategia (Funcion defining trade criteria), no es llamada desde las funciones especiales, pero si desde la función que lleva la contabilidad. L función “Data” es llamada desde la función especial deinit() y, si es necesario, también es llamada desde la función que procesa los errores, o la función que maneja la contabilidad o también la función que procesa los eventos. El archivo que contiene las variables compartidas por todas las funciones, y que está en la cabecera, no es llamado desde ninguna función, puesto que no contiene ninguna función que pueda ser llamada o ser ejecutada. Este archivo solo contiene las declaraciones de las variables globales o compartidas, por esta razón solo es una pequeña parte del EA. Para entender como es que se relacionan las diferentes partes de un EA, vamos a ver como se incluyen y crean estos archivos y en que orden.

Usando archivos externos con “include”
Si un programa contiene gran cantidad de líneas de código, es difícil encontrar y eliminar los errores. El programador tiene que desplazarse por el código muchas veces para hacer retoques al código en una o en otra parte. En estos casos es muy conveniente y cómodo dividir el código en partes, cada una como un archivo separado. En estos archivos separados se puede colocar cualquier parte del código del programa.  Es común que cada función este separada en un archivo diferente. Si varias funciones se interconectan lógicamente, un archivo incluido puede tener la descripción de todas las funciones definidas por el usuario. 
En la sección “información de una cuenta” vimos un código de ejemplo (check.mq4) que protegía del uso desautorizado de un programa. En el asesor experto check.mq4 vimos una función que se encargaba de esa protección y que se llamaba Check(). En el código de un EA que vamos a ver más adelante (usualexpert.mq4), vamos a volver a usar la función Check(), pero esta vez la vamos a colocar en un archivo aparte (Check.mqh) y luego la vamos a incluir en el código principal. 
Función definida por el usuario Check()

bool Check()
Esta función devuelve TRUE si las condiciones de protección se cumplen, si no devuelve FALSE. Las condiciones para devolver TRUE son:

· Si el programa es usado en una cuenta demo;

· Si la cuenta se abre con SuperBank;

· Si el usuario a colocado el código correcto en la variable externa Parol.

El siguiente es el archivo Check.mqh que contiene la descripción del funcionamiento de la función Check(); 
//----------------------------------------------------------------------------------

// Check.mqh

// Este programa esta pensado para servir de ejemplo en el tutorial de MQL4.

//----------------------------------------------------------------------------- 1 --

// Función que comprueba si es legal usar el programa

// Entradas:

// - variable global 'Parol'

// - constante local "SuperBank"

// Devuelve los valores:

// true – si las condiciones se cumplen 

// false - si las condiciones son violadas 

//----------------------------------------------------------------------------- 2 --

extern int Parol=12345; // Contraseña con la que trabaja el programa real
//----------------------------------------------------------------------------- 3 --

bool Check() // Función definida por el usuario
{

if (IsDemo()==true) // Si la cuenta es demo
return(true); // .. entonces no hay limintaciones

if (AccountCompany()=="SuperBank") // Para los clientes corporativos…
return(true); // …no se requiere contraseña
int Key=AccountNumber()*2+1000001; // Calcular la clave

if (Parol==Key) // Si la contraseña esta bien..

return(true); // ..entonces se puede usar en una cuenta real. 
Inform(14); // Envía un informe desautorizando el uso
return(false); // Sale de la función. 
}

//----------------------------------------------------------------------------- 4 –
Es fácil ver que el nombre del archivo Check.mqh es igual que el de la función Check(). Esto no es una regla de MQL4. No es necesario que sean iguales y más si nos damos cuenta que un archivo “.mqh” puede tener varias funciones a dentro. Sin embargo, es muy práctico colocar el mismo nombre a la función y al archivo.  Esto facilita enormemente el trabajo del programador. Usando el nombre de la función, sabrá que el código de esta estará en la ruta “…\experts\include\” con el mismo nombre. Para incluir un archivo externo “.mqh” debemos utilizar la directiva o palabra clave #include.
Directiva #include
#include <Nombre del archivo>
#include " Nombre del archivo"
La directiva #include se puede colocar en cualquier parte del programa. Sin embargo por legibilidad, se debe colocar al inicio del programa. El pre compilador sustituirá #include < Nombre del archivo > (o #include "Nombre del archivo ") por el código del archivo que tenga ese nombre. 

Los corchetes menores/mayores <> significan que el archivo será tomado del directorio predeterminado Terminal_directory “…\experts\include\”. Si el nombre de archivo se coloca en “comillas”, será buscado en el directorio actual, a saber, en el directorio que contiene el archivo principal. 
Abajo esta un asesor experto normal, usualexpert.mq4. Todos los archivos incluidos o importados, se colocan en la parte principal del programa. 

//----------------------------------------------------------------------------------------

// usualexpert.mq4

// Este código solo debe ser usado de manera educacional.
 //----------------------------------------------------------------------------------- 1 --

#property copyright "Copyright © Book, 2007"

#property link "http://AutoGraf.dp.ua"

//----------------------------------------------------------------------------------- 2 --

#include <stdlib.mqh>

#include <stderror.mqh>

#include <WinUser32.mqh>

//----------------------------------------------------------------------------------- 3 --

#include <Variables.mqh> // Variables con descripción 
#include <Check.mqh> // Verifica la legalidad del uso
#include <Terminal.mqh> // Función de la contabilidad
#include <Events.mqh> // Seguimiento de los eventos
#include <Inform.mqh> // Función de manejo de datos 
#include <Trade.mqh> // Función de operaciones
#include <Open_Ord.mqh> // Apertura de ordenes
#include <Close_All.mqh> // Cierre de ordenes
#include <Tral_Stop.mqh> // Modificación de StopLoss 
#include <Lot.mqh> // Calculo de numero de lotes
#include <Criterion.mqh> // Lógica de la estrategia
#include <Errors.mqh> // Procesamiento de Errores
//----------------------------------------------------------------------------------- 4 --

int init() // Función especial 'init'

{

Level_old=MarketInfo(Symbol(),MODE_STOPLEVEL );//Distancia mínima
Terminal(); // Verifica la contabilidad de la cuenta
return; // Salida de init()

}

//----------------------------------------------------------------------------------- 5 --

int start() // Función especial 'start'

{

if(Check()==false) // Si no cumple las condiciones...

return; // ..termina la ejecución. 
PlaySound("tick.wav"); // Ejecuta un sonido en cada tic
Terminal(); // Lleva la contabilidad 
Events(); // Información sobre los eventos
Trade(Criterion()); // Ejecuta un trade si se cumplen los criterios
Inform(0); // Cambia el color de los textos que llevan el informe
return; // Sale de start()

}

//----------------------------------------------------------------------------------- 6 --

int deinit() // Función especial deinit()

{

Inform(-1); // Elimina todos los objetos gráficos
return; // Sale de deinit()

}

//----------------------------------------------------------------------------------- 7 –
En el bloque 2 – 3 incluimos los archivos predeterminados stdlib.mqh, stderror.mqh y WinUser32.mqh usando la directiva #include. Estos archivos no son indispensables, sin embargo son una buena ayuda. Por ejemplo, el archivo stderror.mqh contiene la definición de los códigos de error, que nos hará mas fácil comprender estos códigos.  Si no se están procesando los errores en un programa, este archivo no será necesario. Pero generalmente es útil y normal incluir estos archivos. 
En el bloque 3 – 4 el programa incluye algunos archivos que contienen funciones definidas por el usuario. En la línea que esta abajo, agregamos la función Check.mqh. 

#include <Check.mqh> // Verifica la legalidad del uso

Uno puede pensar que un código como el anterior (usualexpert.mq4) es interpretado por el compilador tal cual como se ve. Sin embargo lo que el pre compilador hace, es sustituir cada línea con la directiva #include, por el código que el archivo tiene. De esta manera, se compila todo como si fuera solo un único archivo “.mq4” que genera un archivo “.ex4”.
En general todos los archivos externos contienen funciones. Sin embargo en el código anterior hay un archivo muy importante que no tiene ninguna función ejecutable. En vez de tener una función, este archivo contiene todas las variables que se utilizaran en el programa, y que se compartirá entre todos los archivos. 

#include <Variables.mqh> 

El código de este archivo es el siguiente:

//----------------------------------------------------------------------------

// Variables.mqh

// Este código solo debe ser usado de manera educacional.

//----------------------------------------------------------------------- 1 --

// Descripción de variables globales

extern double Lots = 0.0;// Cantidad de lotes

extern int Percent = 0; // Porcentaje de fondos asignados

extern int StopLoss =100; // StopLoss para las nuevas órdenes en pips

extern int TakeProfit =40; // TakeProfit para las nuevas órdenes en pips

extern int TralingStop=100; // TralingStop para ordenes de mercado en pips

//----------------------------------------------------------------------- 2 --

int

Level_new, // Nuevo valor para la mínima distancia

Level_old, // Previo valor para la mínima distancia 

Mas_Tip[6]; // Array con tipos de ordenes

// [] order type: 0=B,1=S,2=BL,3=SL,4=BS,5=SS

//----------------------------------------------------------------------- 3 --

double

Lots_New, // Cantidad de lotes para las nueva ordenes
Mas_Ord_New[31][9], // Array con el listado de ordenes abiertas.

Mas_Ord_Old[31][9]; // .. Array con el listado de ordenes abiertas previas.

// 1st index = Numero de ordenes abiertas
// [][0] Sin datos
// [][1] Precio de apertura de la orden
// [][2] StopLoss de la orden
// [][3] TakeProfit de la orden
// [][4] Numero de la orden
// [][5] Volumen de la orden 
// [][6] Tipo de orden 0=B,1=S,2=BL,3=SL,4=BS,5=SS

// [][7] Número mágico de la orden
// [][8] 0/1 para saber si la orden tiene comentarios 
//----------------------------------------------------------------------- 4 –
Según las reglas que maneja MQL4, las variables globales se deben declarar al inicio del código. Por esta razón, el archivo Variables.mqh se incluye al inicio del programa, y está situado por encima de los archivos que contienen las funciones que utilizaran estas variables. 

En algunos casos (raros), es técnicamente posible declarar una variable global en un archivo incluido que contenga una función, en la cual el valor de esta variable primero se utiliza en el programa. En tales casos, es necesario guardar el archivo respetando la regla de posicionamiento de las variables.
En otros casos, es incluso técnicamente imposible hacerlo. Por ejemplo, si tenemos dos archivos incluidos, cada uno de ellos con dos variables globales, una de las cuales se declara en un archivo y la otra se declara en el otro archivo, entonces se obtendrá un error al compilar un programa, ya que, independientemente del orden, una de las variables se utiliza antes de que se declare en el programa. Esta es la razón por la cual, es una práctica habitual y recomendada en los programas normales, declarar todas las variables globales, sin excepción alguna, en un único archivo, y incluirlo en el programa, antes que los otros ficheros que contengan funciones definidas por el usuario.
En el bloque 1-2 del archivo Variables.mqh, las variables externas que se declaran son las que determinan la cantidad de lotes para nuevas órdenes, también hay una variable que determina el porcentaje asignado para nuevos pedidos, también está el stoploss, el TakeProfit y el TralingStop o stop fluctuante. En el bloque 2-4, se declaran otras variables globales que se explicaran y quedaran claras más adelante. Cada archivo externo será explicado paso a paso, en los capítulos subsiguientes. 
